

UNIVERSITATEA "ALEXANDRU-IOAN CUZA" DIN IAȘI

FACULTATEA DE INFORMATICĂ



LUCRARE DE LICENȚĂ

**Verificarea problemei de selecție a activităților
cu profit maxim în Dafny**

propusă de

Roxana Mihaela Timon

Sesiunea: iulie, 2024

Coordonator științific

Conf. Dr. Ciobâcă Ștefan

UNIVERSITATEA "ALEXANDRU-IOAN CUZA" DIN IAȘI

FACULTATEA DE INFORMATICĂ

**Verificarea problemei de selecție a
activităților cu profit maxim în Dafny**

Roxana Mihaela Timon

Sesiunea: iulie, 2024

Coordonator științific

Conf. Dr. Ciobâcă Ștefan

Avizat,
Îndrumător lucrare de licență,
Conf. Dr. Ciobâcă Ștefan.

Data: Semnătura:

Declarație privind originalitatea conținutului lucrării de licență

Subsemnatul **Timon Roxana Mihaela** domiciliat în **România, jud. Vaslui, sat. Valea-Grecului, str. Bisericii, nr. 24**, născut la data de **09 iulie 2000**, identificat prin CNP **6000709375208**, absolvent al Facultății de informatică, **Facultatea de informatică** specializarea **informatică**, promoția 2022, declar pe propria răspundere cunoscând consecințele falsului în declarații în sensul art. 326 din Noul Cod Penal și dispozițiile Legii Educației Naționale nr. 1/2011 art. 143 al. 4 și 5 referitoare la plagiat, că lucrarea de licență cu titlul **Verificarea problemei de selecție a activităților cu profit maxim în Dafny** elaborată sub îndrumarea domnului **Conf. Dr. Ciobâcă Ștefan**, pe care urmează să o susțin în fața comisiei este originală, îmi aparține și îmi asum conținutul său în întregime.

De asemenea, declar că sunt de acord ca lucrarea mea de licență să fie verificată prin orice modalitate legală pentru confirmarea originalității, consimțind inclusiv la introducerea conținutului ei într-o bază de date în acest scop.

Am luat la cunoștință despre faptul că este interzisă comercializarea de lucrări științifice în vederea facilitării falsificării de către cumpărător a calității de autor al unei lucrări de licență, de diplomă sau de disertație și în acest sens, declar pe proprie răspundere că lucrarea de față nu a fost copiată ci reprezintă rodul cercetării pe care am întreprins-o.

Data:

Semnătura:

Declarație de consimțământ

Prin prezenta declar că sunt de acord ca lucrarea de licență cu titlul **Verificarea problemei de selecție a activităților cu profit maxim în Dafny**, codul sursă al programelor și celelalte conținuturi (grafice, multimedia, date de test, etc.) care însoțesc această lucrare să fie utilizate în cadrul Facultății de informatică.

De asemenea, sunt de acord ca Facultatea de informatică de la Universitatea "Alexandru-Ioan Cuza" din Iași, să utilizeze, modifice, reproducă și să distribuie în scopuri necomerciale programele-calculator, format executabil și sursă, realizate de mine în cadrul prezentei lucrări de licență.

Absolvent **Roxana Mihaela Timon**

Data:

Semnătura:

Cuprins

Motivație	2
Intenție	3
Introducere	4
1 Dafny	5
1.1 Prezentare generală	5
2 Programarea dinamică	6
2.1 Avantajele programării dinamice față de celelalte tehnici de proiectare .	6
2.2 Ce este problema de selecție a activităților cu profit maxim ?	7
2.3 Pseudocod	7
2.4 Cum funcționează programarea dinamică în cazul problemei de selecție a activitatilor	9
2.4.1 Soluția parțială și soluția parțială optimă	9
2.4.2 Proprietatea de substructură optimă în cazul problemei de selecție a activităților	10
3 Verificarea problemei de selecție a activităților în Dafny	11
3.1 Date de intrare și predicate specifice	11
3.2 Variabile folosite pentru rezolvare	13
3.3 Punctul de intrare în algoritm	15
3.4 Detalii de implementare	19
3.4.1 Precondiții, postcondițiilor și invariante	21
3.4.2 Funcții importante folosite	23
3.5 Date de ieșire și predicate specifice	24
3.5.1 Descrierea datelor de ieșire	24

3.6	Leme importante in demonstrarea corectitudinii	25
3.7	Mod de lucru	31
3.7.1	Timeout	32
	Concluzii	33
	Bibliografie	34

Motivație

Am ales sa fac această temă deoarece Dafny era un limbaj de programare nou pentru mine si am considerat a fi o provocare. Știam doar că acesta este folosit pentru a asigura o mai mare siguranță si corectitudine, putând fi aplicat in industria aerospațiala, în industria medicala, la dezvoltarea sistemelor financiare, în securitate și criptografie. Totodată, prin demonstrarea corectitudinii unui algoritm in Dafny puteam să-mi folosesc pe lângă cunostințele informatice si pe cele de matematică, de care am fost mereu atrasă. Pentru a crește gradul de complexitate al lucrării am decis sa folosesc ca și tehnică de proiectare a aloritmilor programarea dinamică. Astfel, având posibilitatea sa înțeleg mai bine cum funcționează programarea dinamică și să demonstrez că, cu ajutorul ei se obține o soluție optimă.

Intenție

În cadrul lucrării voi discuta despre limbajul de programare Dafny, surprinzând particularitățile sale, despre problema de selecție a activităților cu profit maxim, despre programarea dinamică și avantajele sale în comparație cu alte tehnici de proiectare a algoritmilor. Voi prezenta, de asemenea, demonstrația de corectitudine a problemei de selecție a activităților cu profit maxim folosind programarea dinamică în Dafny.

Introducere

Lucrarea este structură în 3 capitole:

- **Dafny.** În acest capitol voi prezenta particularitățile limbajului de programare Dafny.
- **Programarea dinamică.** În acest capitol voi reaminti despre programarea dinamică ca tehnică de proiectare a algoritmilor, despre avantajele sale în comparație cu alte tehnici de proiectare, precum Greedy.
- **Problema de selecție a activităților cu profit maxim.** Acest capitol conține prezentarea algoritmului de selecție a activităților cu profit maxim, demonstrația de corectitudine cu ajutorul limbajului de programare Dafny și tehnica de lucru abordată.

Capitolul 1

Dafny

1.1 Prezentare generală

Dafny este un limbaj imperativ de nivel înalt cu suport pentru programarea orientată pe obiecte. Metodele realizate în Dafny au precondiții, postcondiții și invarianți care sunt verificate la compilare, bazându-se pe soluționatorul SMT Z3. În cazul în care o postcondiție nu poate fi stabilită (fie din cauza unui timeout, fie din cauza faptului că aceasta nu este valabilă), compilarea eșuează. Prin urmare, putem avea un grad ridicat de încredere într-un program verificat cu ajutorul sistemului Dafny. Acesta a fost conceput pentru a facilita scrierea unui cod corect, în sensul de a nu avea erori de execuție, dar și corect în sensul de a face ceea ce programatorul a intenționat să facă.[1] Dafny, ca platforma de verificare, permite programatorului să specifice comportamentul dorit al programelor sale, astfel încât acesta să verifice dacă implementarea efectivă este conformă cu acel comportament. Cu toate acestea, acest proces nu este complet automat iar uneori programatorul trebuie să ghideze verificatorul.[2]

Capitolul 2

Programarea dinamică

Programarea dinamică este o tehnică de proiectare a algoritmilor utilizată pentru rezolvarea problemelor de optimizare. Pentru a rezolva o anumită problemă folosind programarea dinamică, trebuie să identificăm în mod convenabil mai multe subprobleme. După ce alegem subproblemele, trebuie să stabilim cum se poate calcula soluția unei subprobleme în funcție de alte subprobleme. Principala idee din spatele programării dinamice constă în stocarea rezultatelor subproblemele pentru a evita recalcularea lor de fiecare dată când sunt necesare. În general, programarea dinamică se aplică pentru probleme de optimizare pentru care algoritmi greedy nu produc în general soluția optimă. [4]

2.1 Avantajele programării dinamice față de celelalte tehnici de proiectare

În ceea ce privește programarea dinamică și tehnica greedy, în ambele cazuri apare noțiunea de subproblemă și proprietatea de substructură optimă. De fapt, tehnica greedy poate fi gândită ca un caz particular de programare dinamică, unde rezolvarea unei probleme este determinată direct de alegerea greedy, nefiind nevoie de a enumera toate alegerile posibile.[4] Avantajele programării dinamice față de tehnica greedy sunt:

- **Optimizare Globală:** : Programarea dinamică are capacitatea de a găsi soluția optimă globală pentru o problemă, în timp ce algoritmi greedy pot fi limitați la luarea deciziilor locale care pot duce la o soluție suboptimală.

- **Flexibilitate:** Programarea dinamică poate fi utilizată pentru o gamă mai largă de probleme, inclusiv cele care implică restricții mai complexe sau soluții care necesită evaluarea mai multor posibilități. În comparație, algoritmi greedy sunt adesea limitați la problemele care pot fi rezolvate prin luarea deciziilor locale în fiecare pas.

Cu toate acestea, algoritmi greedy pot fi mai potriviți pentru problemele care permit luarea de decizii locale și producerea rapidă a unei soluții aproximative.

2.2 Ce este problema de selecție a activităților cu profit maxim ?

Problema de selecție a activităților cu profit maxim este o problemă de optimizare care returnează pentru o listă de activități distincte (diferă prin cel puțin un timp) caracterizate prin timp de început, timp de încheiere și profit, ordonate după timpul de încheiere, o secvență de activități care nu se suprapun două câte două și care produc un profit maxim.

```
Date de intrare: O secvență de activități caracterizate prin
{ timp de început , timp de încheiere , profit}

Activitate 1: {1, 2, 50}
Activitate 2: {3, 5, 20}
Activitate 3: {6, 19, 100}
Activitate 4: {2, 100, 200}

Date de ieșire: Profitul maxim este 250, pentru soluția optim
formată din activitatea 1 și activitatea 4.
```

2.3 Pseudocod

```
1
2 struct Activitate {
3     timp de început: int
4     timp de încheiere : int
5     profit : int
6 }
7 function planificareActivitatiPonderate(activitati):
8     // activitățile sunt sortate crescător după timpul de încheiere
```

```

9      // initializam un vector pentru a stoca profiturile maxime pentru
    fiecare activitate, fie dp un vector de dimensiune n, unde n este
    numarul de activitati
10     dp[0] = activitati[0].profit
11     solutie[0] = [activitati[0]] //un vector binar 0 - nu am ales
    activitatea si 1 - am ales activitatea
12     solutiiOptime = solutie //stocam solutiile optime la fiecare pas
13     // Programare dinamica pentru a gasi profitul maxim
14     pentru i de la 1 la n-1:
15         // Gaseste cea mai recenta activitate care nu intra in conflict cu
    activitatea curenta
16         solutiaCuActivitateaI = [1] // selectam activitatea curenta
17         ultimaActivitateNeconflictuala =
    gasesteUltimaActivitateNeconflictuala(activitati, i)
18
19         // Calculeaz profitul maxim incluzand activitatea curenta si
    excluzand-o
20         profitulCuActivitateaCurenta = activitati[i].profit
21         daca ultimaActivitateNeconflictuala != -1:
22             profitulIncluderiiActivitatiiCurente += dp[
    ultimaActivitateNeconflictuala]
23             solutiaCuActivitateaI = solutiiOptime[
    ultimaActivitateNeconflictuala] + solutiaCuActivitateaI
24             dp[i] = maxim(profitulIncluderiiActivitatiiCurente, dp[i-1])
25
26         daca profitulIncluderiiActivitatiiCurente > dp[i-1]:
27             solutie = solutieCuActivitateaI
28         else
29             solutie = solutie + [0]
30             solutiiOptime = solutiiOptime + solutie
31     // Returneaza solutia si profitul maxim
32     return solutie, dp[n-1]
33
34 function gasesteUltimaActivitateNeconflictuala(activitati, indexCurent):
35     pentru j de la indexCurent-1 la 0:
36         daca activitati[j].timpDeIncheiere <= activitati[indexCurent].
    timpDeInceput:
37             return j
38     return -1

```

2.4 Cum funcționează programarea dinamică în cazul problemei de selecție a activităților

Pentru problema de selecție a activităților se poate folosi programarea dinamică, deoarece aceasta poate fi împărțită în subprobleme, respectiv la fiecare pas putem forma o soluție parțială optimă, de al cărei profit ne putem folosi la următorul pas.

2.4.1 Soluția parțială și soluția parțială optimă

O soluție parțială trebuie să conțină activități care nu se suprapun și să aibă o lungime prestabilită. O soluție parțială este și optimă, dacă profitul acesteia este mai mare sau egal decât al oricărei alte soluții parțiale de aceeași lungime.

Pentru datele de intrare precizate de mai sus se obțin următoarele valori:

La primul pas:

1. soluția parțială optimă este formată din Activitatea 1.
2. iar profit-ul maxim este 50.

La al 2-lea pas:

1. deoarece Activitatea 1 nu se suprapune cu Activitatea 2 se obține soluția parțială optimă formată din Activitatea 1 și Activitatea 2.
2. profitul optim la pasul 2 este $50 + 20 = 70$, care este mai mare decât cel anterior (condiție necesară).

La al 3-lea pas:

1. soluția parțială optimă este formată din Activitatea 1, Activitatea 2 și Activitatea 3, deoarece Activitatea 3 nu se suprapune cu activitatea 2 (înseamnă că nu se suprapune cu soluția parțială de la al 2-lea pas, fiind ordonate după timpul se sfârșit) și le putem concatena.
2. profitul pentru această soluție parțială este strict mai mare decât cel de la pasul 2, noul profit optim devenind 170.

La al 4-lea pas:

1. soluția parțială ce conține Activitatea 4 este formată din Activitatea 4 și Activitatea 1.
2. profitul pentru această soluție parțială este strict mai mare decât profitul optim anterior = 170, noul profit optim devenind 250.

Astfel, soluția problemei este cea de la ultimul pas, fiind formată din Activitatea 1 și Activitatea 4 și având profitul optim 250.

2.4.2 Proprietatea de substructură optimă în cazul problemei de selecție a activităților

În cazul problemei de selecție a activităților, proprietatea de substructură optimă este identificată atunci când eliminăm o activitate dintr-o soluție parțială optimă și obținem tot o soluție parțială optimă. Altfel, dacă această proprietate nu ar fi valabilă, înseamnă că soluția parțială optimă pe care o descompunem nu este cu adevărat optimă.

Capitolul 3

Verificarea problemei de selecție a activităților in Dafny

3.1 Date de intrare și predicate specifice

Pentru a reprezenta o activitate am folosit un tuplu.

```
datatype Job = Tuple(jobStart: int, jobEnd: int, profit: int)
```

Problema de selectie a activităților are ca date de intrare un singur parametru:

- **jobs** de tipul *seq*<Job> : reprezintă secvența de activități

Un predicat este o funcție care returnează o valoare booleana.[3] Acestea sunt folosite ca precondiții și postcondiții în metode. Predicatele pe care le-am folosit la validarea datelor de intrare sunt:

```
predicate validJob(job: Job)
{
    job.jobStart < job.jobEnd && job.profit >= 0
}
```

```
predicate validJobsSeq(jobs: seq<Job>)
{
    forall job :: job in jobs ==> validJob(job)
}
```

Predicatul **validJobsSeq(jobs: seq<Job>)** asigură faptul că secvența de activități primită ca dată de intrare conține doar activități valide, ceea ce înseamnă că timpul de

început este anterior timpului de încheiere și că profitul asociat fiecărei activități este un număr pozitiv.

```
predicate JobComparator(job1: Job, job2: Job)
{
    job1.jobEnd <= job2.jobEnd
}

predicate sortedByActEnd(s: seq<Job>)
    requires validJobsSeq(s)
{
    forall i, j :: 0 <= i < j < |s| ==> JobComparator(s[i], s[j])
}
```

Predicatul **sortedByActEnd(s: seq<Job>)** ne asigură că în secvența de activități primită activitățile sunt sortate din punct de vedere al timpului de încheiere.

```
predicate distinctJobs(j1: Job, j2: Job)
    requires validJob(j1) && validJob(j2)
{
    j1.jobStart != j2.jobStart || j1.jobEnd != j2.jobEnd
}
```

```
predicate distinctJobsSeq(s: seq<Job>)
    requires validJobsSeq(s)
{
    forall i, j :: 0 <= i < j < |s| ==> distinctJobs(s[i], s[j])
}
```

Pentru a ne asigura că secvența de activități primită ca input nu conține activități identice am folosit predicatul **distinctJobsSeq(s: seq<Job>)**.

Aceste 3 predicate le-am unit într-unul singur, **validProblem**.

```
predicate validProblem(jobs: seq<Job>)
{
    1 <= |jobs| && validJobsSeq(jobs) && sortedByActEnd(jobs)
    && distinctJobsSeq(jobs)
}
```

3.2 Variabile folosite pentru rezolvare

- **solution** de tipul $seq<int>$: reprezintă soluția parțială optimă obținută la fiecare iteratie
- **partialSol** de tipul $seq<int>$: reprezintă o soluție parțială
- **dp** de tipul $seq<int>$: ce conține toate profiturile optime obținute la fiecare pas
- **allSol** de tipul $seq<seq<int>>$: reprezintă soluțiile parțiale optime obținute la fiecare pas

Predicatele pe care le-am folosit pentru a verifica corectitudinea valorilor variabilelor intermediare folosite sunt:

```
predicate overlappingJobs (j1:Job, j2:Job)
```

```
  requires validJob(j1)
```

```
  requires validJob(j2)
```

```
{
```

```
  j1.jobEnd > j2.jobStart && j2.jobEnd > j1.jobStart
```

```
}
```

```
predicate hasNoOverlappingJobs (partialSol: seq<int>, jobs: seq<Job>)
```

```
  requires validJobsSeq(jobs)
```

```
{
```

```
  |partialSol| <= |jobs| && forall i, j :: 0 <= i < j < |partialSol|
```

```
    ==> (partialSol[i] == 1 && partialSol[j] == 1)
```

```
    ==> !overlappingJobs(jobs[i], jobs[j])
```

```
}
```

```
predicate isPartialSol (partialSol: seq<int>, jobs: seq<Job>, length: int)
```

```
  requires validJobsSeq(jobs)
```

```
{
```

```
  |partialSol| == length && forall i :: 0 <= i <= |partialSol| - 1 ==>
```

```
    (0 <= partialSol[i] <= 1) && hasNoOverlappingJobs(partialSol, jobs)
```

```
}
```

Cu ajutorul acestui predicat, **isPartialSol(partialSol: seq<int>, jobs: seq<Job>, length: int)** am verificat ca o soluție parțială să aibă lungimea dorită, să conțină doar

valori de 0 și 1, iar activitățile care corespund acestui vector caracteristic să nu se suprapună.

```

predicate HasLessProf(partialSol: seq<int>, jobs: seq<Job>, maxProfit: int,
    position: int)
    requires validJobsSeq(jobs)
    requires 0 <= position < |partialSol| <= |jobs|
{
    PartialSolProfit(partialSol, jobs, position) <= maxProfit
}

ghost predicate isOptParSol(partialSol: seq<int>, jobs: seq<Job>,
    length: int)
    requires validJobsSeq(jobs)
    requires 1 <= |jobs|
    requires length == |partialSol|
    requires 1 <= |partialSol| <= |jobs|
{
    isPartialSol(partialSol, jobs, length) &&
    forall otherSol :: isPartialSol(otherSol, jobs, length)
        ==> HasLessProf(otherSol, jobs,
            PartialSolProfit(partialSol, jobs, 0), 0)
}

```

Predicatul **isOptParSol(partialSol: seq<int>, jobs: seq<Job>, length: int)** are rolul de a verifica dacă o soluție parțială este și optimă. Pentru ca aceasta să fie optimă trebuie să îndeplinească următoarele condiții:

1. să fie o soluție parțială
2. orice altă soluție parțială are un profit mai mic decât soluția parțială optimă.

```

ghost predicate isOptParSolDP(partialSol: seq<int>, jobs: seq<Job>,
    length : int, dp:int)
    requires validJobsSeq(jobs)
    requires 1 <= |partialSol|
    requires 1 <= length <= |jobs|
{

```

```

|partialSol| == length && isOptParSol(partialSol, jobs, length)
&& HasProfit(partialSol, jobs, 0, dp)
}

ghost predicate OptParSolutions(allSol: seq<seq<int>>, jobs: seq<Job>,
dp:seq<int>, index: int)
  requires validJobsSeq(jobs)
  requires |dp| == |allSol| == index
  requires 1 <= index <= |jobs|
{
  forall i : int :: 0 <= i < index ==> |allSol[i]| == i + 1
  && isOptParSolDP(allSol[i], jobs, i + 1, dp[i])
}

```

Pentru a putea verifica dacă allSol conține doar soluții parțiale optime, am folosit predicatul **OptParSolutions(allSol: seq<seq<int>>, jobs: seq<Job>, dp:seq<int>, index: int)**. Acest predicat verifică pentru o secvență de secvențe:

1. fiecare secvență să aibă lungimea dorită
2. fiecare secvență să fie o soluție parțială optimă
3. fiecare secvență să aibă profitul dorit (optim)

3.3 Punctul de intrare în algoritm

```

method WeightedJobScheduling(jobs: seq<Job>) returns (sol: seq<int>,
profit : int)
  requires validProblem(jobs)
  ensures isSolution(sol, jobs)
  ensures isOptimalSolution(sol, jobs)
{
  var dp :seq<int> := [];
  var dp0 := jobs[0].profit;
  dp := dp + [dp0];
  var solution : seq<int> := [1];
  var i: int := 1;

```

```

var allSol : seq<seq<int>> := [];
allSol := allSol + [[1]];

assert |solution| == 1;
assert |allSol[0]| == |solution|;
assert 0 <= solution[0] <= 1;

assert isPartialSol(solution, jobs, i);
assert validJob(jobs[0]); //profit >=0
assert isOptParSol(solution, jobs, i);

while i < |jobs|
  invariant 1 <= i <= |jobs|
  decreases |jobs| - i
  invariant i == |dp|
  invariant 1 <= |dp| <= |jobs|
  decreases |jobs| - |dp|
  invariant isPartialSol(solution, jobs, i)
  invariant |solution| == i
  invariant i == |allSol|
  decreases |jobs| - |allSol|
  decreases |jobs| - |allSol[i-1]|
  invariant isPartialSol(allSol[i-1], jobs, i)
  invariant HasProfit(solution, jobs, 0, dp[i - 1])
  invariant HasProfit(allSol[i - 1], jobs, 0 , dp[i - 1])
  invariant OptParSolutions(allSol, jobs, dp, i)
  invariant isOptParSol(allSol[i - 1], jobs, i)
  invariant forall partialSol :: |partialSol| == i
    && isPartialSol(partialSol, jobs, i) ==>
      HasLessProf(partialSol, jobs, dp[i - 1], 0)
  invariant isOptParSol(solution, jobs, i)
{
  var maxProfit, partialSolWithI := MaxProfitWithJobI(jobs, i, dp, allSol);

  if dp[i-1] >= maxProfit
  {

```

```

    solution, dp :=
        leadsToOptWithoutJobI(jobs, dp, allSol, i, maxProfit, solution);
    assert isOptParSol(solution, jobs, i + 1);

}

else
{

    solution, dp :=
        leadsToOptWithJobI(jobs, dp, allSol, i, maxProfit, partialSolWithI);
    assert isOptParSol(solution, jobs, i + 1);

}
allSol := allSol + [solution];
i := i + 1;
}

sol := solution;
profit := dp[|dp|-1];
}

```

Metoda care îmi generează soluția parțială ce conține activitatea i , este **MaxProfitWithJobI**. În Dafny, metodele sunt subrutine care pot fi executate în timpul execuției și pot avea efecte secundare, cum ar fi modificarea obiectelor heap.[2] Această metodă primește ca parametri:

1. **jobs** : secvența de activități (problema)
2. **i** : poziția (din secvența de activități primită ca dată de intrare) pe care se afla activitatea cu care vrem să formăm o soluție parțială
3. **dp** : secvența cu profiturile optime obținute la pașii anteriori
4. **allSol**: secvența cu soluțiile parțiale optime obținute la pașii anteriori

Și returnează:

1. **maxProfit**: profitul pentru soluția parțială ce conține activitatea de pe poziția i din secvența de activități primită ca dată de intrare

2. optParSolWithI: soluția parțială optimă ce conține activitatea de pe poziția i

```

method MaxProfitWithJobI(jobs: seq<Job>, i: int, dp: seq<int>,
allSol :seq<seq<int>>) returns (maxProfit:int, optParSolWithI: seq<int>)
  requires validProblem(jobs)
  requires PositiveProfitsDP(dp)
  requires 1 <= i < |jobs|
  requires |allSol| == i
  requires |dp| == i
  requires OptParSolutions(allSol, jobs, dp, i)
  ensures isPartialSol(optParSolWithI, jobs, i + 1)
  ensures maxProfit == PartialSolProfit(optParSolWithI, jobs, 0)
  ensures partialSolutionWithJobI(optParSolWithI, jobs, i)
  ensures forall partialSol :: |partialSol| == i + 1 &&
    partialSolutionWithJobI(partialSol, jobs, i) ==>
      HasLessProf(partialSol, jobs, maxProfit, 0)
{

  var max_profit := 0;
  var partialSolutionPrefix : seq<int> := [];
  var partialSolution : seq<int> := [];
  var j := i - 1;
  var length := 0;

  while j >= 0 && jobs[j].jobEnd > jobs[i].jobStart
    invariant - 1 <= j < i
    invariant forall k :: j < k < i ==> jobs[k].jobEnd > jobs[i].jobStart
    invariant forall k :: j < k < i ==> validJob(jobs[k])
    invariant forall k :: j < k < i ==> JobComparator(jobs[k], jobs[i])
    invariant forall k :: j < k < i ==> jobs[k].jobEnd > jobs[k].jobStart
    invariant forall k :: j < k < i ==> overlappingJobs(jobs[k], jobs[i])
  {
    j := j - 1;
  }

  assert j != -1 ==> !overlappingJobs(jobs[j], jobs[i]);

```

```

if j >= 0
{
    max_profit, partialSolution, length :=
        OptParSolWhenNonOverlapJob(jobs, i, dp, allSol, j);
}
else
{
    max_profit, partialSolution, length :=
        OptParSolWhenOverlapJob(jobs, i, dp);
}
assert isPartialSol(partialSolution, jobs, length);
assert forall partialSol :: |partialSol| == i + 1
    && partialSolutionWithJobI(partialSol, jobs, i)
        ==> HasLessProf(partialSol, jobs, max_profit, 0) ;
maxProfit := max_profit;
optParSolWithI := partialSolution;
}

```

3.4 Detalii de implementare

Cum functioneaza algoritmul:

1. **Date de intrare:** primim ca date de intrare variabila activităţi care reprezinta o secvenţă de activităţi caracterizate prin timp de inceput, timp de incheiere si profit, distincte (difere prin cel putin un timp), sortate dupa timpul de incheiere

2. **Programare dinamica:**

La fiecare iteratie stocam solutiile optime ale subproblemelor si profiturile acestora.

- variabila solution, care reprezinta soluţia parţială optimă la fiecare pas, respectiv soluţia optimă finala
- variabila dp, care va conţine profiturile optime obtinute la fiecare iteratie
- variabila allSol care va conţine toate solutiile partiale optime obtinute la fiecare pas

3. **Iteratii:**

- **La prima iteratie** $\text{solution} = [1]$, $\text{dp} = \text{activități}[1].\text{profit}$, $\text{allSol} = [[1]]$
- **Selectăm activitatea de pe poziția i**, unde $i = 1 \dots |\text{activități}| - 1$, în ordinea în care au fost declarate în secvența de intrare
- **Formăm soluția parțială ce conține activitatea de pe poziția i**
 - **profitul** pentru soluția parțială cu activitatea i are valoarea initială **activități[i].profit**
 - **cautăm o activitate j**, unde $0 \leq j < i$, care nu se suprapune cu activitatea i, adică $\text{activități}[j].\text{timpDeÎnceput} \leq \text{activități}[i].\text{timpDeÎnceput}$
 - **dacă $j \geq 0$** atunci soluția parțială cu activitatea de pe poziția i va fi formată din $\text{allSol}[j] + 0 \dots 0 + 1$, unde:
 - * $\text{allSol}[j]$ conține soluția parțială optimă cu activitățile de până la poziția j inclusiv
 - * 0-urile reprezintă, **dacă există**, activitățile dintre j și i care se suprapun cu activitatea de pe poziția i
 - * 1 - înseamnă că activitatea i este selectată
 - iar **profitul** pentru soluția parțială ce conține activitatea i este egal cu **$\text{dp}[j] + \text{activități}[i].\text{profit}$**
 - **dacă $j == -1$** atunci soluția parțială cu activitatea de pe poziția i va fi formată din $0 \dots 0 + 1$, unde:
 - * 0-urile reprezintă, activitățile din fața activității i care se suprapun cu aceasta.
 - * 1 - înseamnă că activitatea de pe poziția i a fost selectată
 - iar **profitul** rămâne egal cu **activități[i].profit**
- **Comparam** profitul soluției parțiale ce conține activitatea de pe poziția i cu profitul care s-ar obține fără activitatea curentă ($\text{dp}[i-1]$) și **actualizăm** valoarea variabilelor dp și solution
 - dacă $\text{dp}[i-1] \geq$ profitul soluției parțiale cu activitatea i, atunci $\text{dp}[i] = \text{dp}[i-1]$ și $\text{solution} = \text{solution} + [0]$
 - dacă $\text{dp}[i-1] <$ profitul soluției parțiale cu activitatea i, atunci $\text{dp}[i] =$ profitul soluției parțiale cu activitatea i și $\text{solution} =$ soluția parțială ce conține activitatea i

- **Actualizam valoarea variabila allSol** care retine solutiile partiale optime obtinute la fiecare iteratie, $\text{allSol} = \text{allSol} + [\text{solution}]$

4. **Datele de ieşire ale problemei** sunt reprezentate de soluţia optimă **sol** si de profitul maxim, $\text{profit} = \text{dp}[|\text{activităţi}| - 1]$

3.4.1 Precondiții, postcondițiilor și invarianți

Metodele și funcțiile Dafny sunt adnotate cu precondiții, postcondiții, invarianți și afirmații (assert-uri), iar verificatorul generează condiții de verificare care sunt trimise unui rezolvator SMT care verifică dacă acestea sunt valide.[2]

Precondițiile reprezintă condiții care trebuie să fie îndeplinite la intrarea într-o metodă sau funcție.

Postcondițiile reprezintă condiții care trebuie să fie îndeplinite la ieșirea dintr-o metodă sau funcție.

Invariantii, la fel ca și precondițiile și postcondițiile, reprezintă condiții care trebuie să fie îndeplinite la intrarea în buclă, în timpul buclei și la ieșirea din aceasta.

În cazul problemei de selecție a activităților precondițiile care trebuie îndeplinite sunt următoarele:

```
method WeightedJobScheduling (jobs: seq<Job>) returns
(sol: seq<int>, profit : int)
    requires validProblem(jobs)
    ensures isSolution(sol, jobs)
    ensures isOptimalSolution(sol, jobs)

predicate validProblem(jobs: seq<Job>)
{
    1 <= |jobs| && validJobsSeq(jobs) && sortedByActEnd(jobs)
    && distinctJobsSeq(jobs)
}
```

Astfel, secvența de activități primită ca input trebuie să conțină doar activități valide, distincte și sortate după timpul de încheiere (problemă validă). La final, când algoritmul se termină, soluția returnată trebuie să îndeplinească următoarele postcondiții: ca aceasta reprezintă o soluție și ca aceasta este optimă.

```

predicate isSolution(solution: seq<int>, jobs: seq <Job>)
  requires validJobsSeq(jobs)
{
  isPartialSol(solution, jobs, |jobs|)
}

```

O soluție înseamnă ca secvența returnată să îndeplinească proprietatea de soluție parțială, dar cu lungimea egală cu cea a secvenței de activități primită ca dată de intrare.

La fiecare iterație a algoritmului, următoarele proprietăți sunt invariante:

```

while i < |jobs|
  invariant 1 <= i <= |jobs|
  decreases |jobs| - i
  invariant i == |dp|
  invariant 1 <= |dp| <= |jobs|
  decreases |jobs| - |dp|
  invariant isPartialSol(solution, jobs, i)
  invariant |solution| == i
  invariant i == |allSol|
  decreases |jobs| - |allSol|
  decreases |jobs| - |allSol[i-1]|
  invariant isPartialSol(allSol[i-1], jobs, i)
  invariant HasProfit(solution, jobs, 0, dp[i - 1])
  invariant HasProfit(allSol[i - 1], jobs, 0, dp[i - 1])
  invariant allSol[i - 1] == solution
  invariant OptParSolutions(allSol, jobs, dp, i)
  invariant isOptParSol(allSol[i - 1], jobs, i)
  invariant forall partialSol :: |partialSol| == i
    && isPartialSol(partialSol, jobs, i) ==>
      HasLessProf(partialSol, jobs, dp[i - 1], 0)
  invariant forall i :: 0 <= i < |dp| ==> dp[i] >= 0
  invariant isOptParSol(solution, jobs, i)

```

Un invariant foarte important este:

```

invariant forall partialSol :: |partialSol| == i
  && isPartialSolution(partialSol, jobs, i) ==>
    HasLessProfit(partialSol, jobs, dp[i - 1], 0)

```

deoarece cu ajutorul lui ne asigurăm că la fiecare pas variabila dp reține profitul optim.

La fel de importante sunt și:

invariant `isPartialSol(solution, jobs, i)`

invariant `isOptParSol(solution, jobs, i)`

care ne asigură că variabila `solution` reține la fiecare pas atât o soluție parțială, cât și o soluție parțială optimă.

3.4.2 Funcții importante folosite

Funcțiile reprezintă un element important pe care le-am folosit la verificarea corectitudinii algoritmului ales. Funcțiile în Dafny sunt asemănătoare funcțiilor matematice, fiind constituite dintr-o singură instrucțiune al cărui tip de return este menționat în antetul acestora. Acestea sunt prezente doar la compilare și nu pot avea efecte secundare.[2] Una din funcțiile pe care am folosit-o cel mai des este funcția: *PartialSolutionPrefixProfit(solution: seq<int>, jobs: seq<Job>, index: int): int* cu ajutorul căreia calculez profitul unei soluții parțiale, după formula

$$\sum_{i=0}^{|partialSol|} partialSol[i] * jobs[i].$$

```
function PartialSolProfit(solution: seq<int>, jobs: seq<Job>, index: int)
    :int
requires 0 <= index <= |solution| <= |jobs|
decreases |solution| - index
ensures PartialSolProfit(solution, jobs, index) ==
    if index == |solution| then 0 else
        solution[index] * jobs[index].profit +
            PartialSolProfit(solution, jobs, index + 1)
{

if index == |solution| then 0 else
    solution[index] * jobs[index].profit +
        PartialSolProfit(solution, jobs, index + 1)
}
```

3.5 Date de ieșire și predicate specifice

- **sol** de tipul *seq<int>* : reprezintă soluția optimă finală pentru secvența de activități primită
- **profit** de tipul *int* : reprezintă profitul maxim al soluției optime finale

```
predicate isSolution(solution: seq<int>, jobs: seq<Job>)  
  requires validJobsSeq(jobs)  
{  
  isPartialSol(solution, jobs, |jobs|)  
}
```

```
ghost predicate isOptimalSolution(solution: seq<int>, jobs: seq<Job>)  
  requires validJobsSeq(jobs)  
  requires 1 <= |jobs|  
  requires |solution| == |jobs|  
{  
  isSolution(solution, jobs) &&  
  forall otherSol :: isSolution(otherSol, jobs) ==>  
    PartialSolProfit(solution, jobs, 0) >= PartialSolProfit(otherSol, jobs, 0)  
}
```

Predicatul **isSolution(solution: seq<int>, jobs: seq<Job>)** este asemănător cu predicatul **isPartialSolution(partialSol: seq<int>, jobs: seq<Job>, length: int)**, doar că după cum se poate observa din antetul acestuia, lipsește variabila *length*, deoarece acum lungimea soluției trebuie să fie egală cu lungimea secvenței de activități primită ca dată de intrare. Același lucru se aplică și pentru predicatul **isOptimalSolution(solution: seq<int>, jobs: seq<Job>)** care verifică:

1. secvența primită să fie o soluție
2. orice altă secvență care este o soluție să aibă un profit mai mic decât aceasta

3.5.1 Descrierea datelor de ieșire

Variabila de ieșire, denumită "sol", este reprezentată printr-un vector caracteristic al secvenței de activități primită ca date de intrare. Acest vector conține doar valori de

0 și 1, unde 0 înseamnă că o activitate nu a fost selectată, în timp ce 1 indică faptul că activitatea respectivă a fost selectată.

3.6 Leme importante în demonstrarea corectitudinii

Unul din punctele complexe în dezvoltarea codului de verificare pentru problema de selecție a activităților a fost atunci când a trebuit să recurg la leme pentru a demonstra anumite proprietăți. Lemele reprezintă blocuri de instrucțiuni care au ca scop demonstrarea unor teoreme pe care Dafny nu reușește să le demonstreze de unul singur. [3]

Lemma **OtherSolHasLessProfitThenMaxProfit2** am folosit-o pentru a demonstra că orice altă **soluție parțială** care:

1. **conține activitatea de pe poziția i**
2. orice activitatea aflată pe oricare din pozițiile $j + 1, \dots, i - 1$ **se suprapune cu activitatea i**, unde $0 \leq j < i$
3. activitatea de pe poziția **j nu se suprapune** cu activitatea de pe poziția **i**

are un **profit mai mic sau egal** decât cel al soluției parțiale care îndeplinește aceleași proprietăți menționate mai sus, doar că pentru aceasta în plus **substructura formată din activitățile de pe poziția 0 până la poziția j este optimă**.

```
lemma OtherSolHasLessProfThenMaxProfit2 (partialSol: seq<int>, jobs : seq<Job>,
i: int, j : int, max_profit : int, allSol : seq<seq<int>>, dp: seq<int>)
  requires validJobsSeq(jobs)
  requires 1 <= |jobs|
  requires 0 <= j < i < |jobs|
  requires |allSol| == |dp| == i
  requires OptParSolutions(allSol, jobs, dp, i)
  requires isOptParSol(allSol[j], jobs, j + 1)
  requires max_profit == dp[j] + jobs[i].profit
  requires forall k :: j < k < i ==> overlappingJobs(jobs[k], jobs[i])
  requires !overlappingJobs(jobs[j], jobs[i])
  requires |partialSol| == i + 1
  requires partialSolutionWithJobI(partialSol, jobs, i)
```

```

requires PartialSolProfit(partialSol, jobs, i) == jobs[i].profit;
requires isPartialSol(partialSol, jobs, i + 1)
requires forall k :: j < k < i ==> partialSol[k] == 0
ensures HasLessProf(partialSol, jobs, max_profit, 0)
{

var k : int := i - 1;
assert |partialSol| == i + 1;
assert j <= k < i;
//assert !exists k' :: k < k' < i;

assert forall k' :: k < k' < i ==> partialSol[k'] == 0;
assert PartialSolProfit(partialSol, jobs, i) == jobs[i].profit;

ComputeProfitWhenOnly0BetweenJI(partialSol, jobs, i, j);
assert PartialSolProfit(partialSol, jobs, j + 1) == jobs[i].profit;

//presupunem contrariul
if !HasLessProf(partialSol, jobs, max_profit, 0)
{
    var profit' := PartialSolProfit(partialSol, jobs, 0);
    assert max_profit == dp[j] + jobs[i].profit;

    assert HasMoreProfit(partialSol, jobs, max_profit, 0);
    assert !HasLessProf(partialSol, jobs, max_profit, 0);

    assert partialSol[..j+1] + partialSol[j+1..] == partialSol;

    SplitSequenceProfitEquality(partialSol, jobs, 0, j + 1);
    EqOfProfitFuncFromIndToEnd(partialSol, jobs, 0);
    EqOfProfFuncUntilIndex(partialSol, jobs, 0, j + 1);
    EqOfProfitFuncFromIndToEnd(partialSol, jobs, j + 1);

    assert PartialSolProfit(partialSol[..j + 1], jobs, 0) +
        PartialSolProfit(partialSol, jobs, j + 1) ==
        PartialSolProfit(partialSol, jobs, 0);

```

```

    assert PartialSolProfit(partialSol, jobs, j + 1) == jobs[i].profit; //(2)

    var partialSol' :seq<int> := partialSol[..j + 1];
    assert isPartialSol(partialSol', jobs, j + 1);
    var profit := PartialSolProfit(partialSol', jobs, 0); //(1)

    assert |partialSol'| == j + 1;
    assert profit + jobs[i].profit == profit';
    assert profit + jobs[i].profit > max_profit;
    assert profit > max_profit - jobs[i].profit;
    assert profit > dp[j];
    HasMoreProfThanOptParSol(allSol[j], jobs, partialSol');
    assert !isOptParSol(allSol[j], jobs, j + 1); //contradictie
    //assume false;
    assert false;
}

assert HasLessProf(partialSol, jobs, max_profit, 0);

}

```

Această lema am folosit-o în cadrul metodei **OptParSolWhenNonOverlapJob** (linia 79) pe care o apelez în metoda **MaxProfitWithJobI** pe ramura cu $j \geq 0$, adică atunci când găsim o activitate pe o poziție j în fața activității de pe poziția i , care nu se suprapune cu aceasta.

```

method OptParSolWhenNonOverlapJob(jobs: seq <Job>, i: int, dp: seq<int>,
allSol :seq<seq<int>>, j : int)
returns (maxProfit:int, partialSolution: seq<int>, length: int)
    requires validProblem(jobs)
    requires 0 <= j < i < |jobs|
    requires |allSol| == i
    requires |dp| == i
    requires OptParSolutions(allSol, jobs, dp, i)
    requires !overlappingJobs(jobs[j], jobs[i]);
    requires jobs[j].jobEnd <= jobs[i].jobStart
    requires forall k :: j < k < i ==> overlappingJobs(jobs[k], jobs[i])

```



```

requires !overlappingJobs(jobs[j], jobs[i]);
ensures isPartialSol(partialSolution, jobs, i + 1)
ensures partialSolutionWithJobI(partialSolution, jobs, i)
ensures maxProfit == PartialSolProfit(partialSolution, jobs, 0)
ensures forall partialSol :: |partialSol| == i + 1 &&
  partialSolutionWithJobI(partialSol, jobs, i) ==>
  HasLessProf(partialSol, jobs, maxProfit, 0)
ensures length == i + 1;
{
  var partialSolutionPrefix : seq<int> := [];
  var max_profit : int := 0 ;
  length := 0;

  partialSolutionPrefix := allSol[j];
  length := length + |allSol[j]|;

  assert forall i :: 0 <= i <= length - 1 ==>
    0 <= partialSolutionPrefix[i] <= 1;
  assert hasNoOverlappingJobs(partialSolutionPrefix, jobs);

  max_profit := max_profit + dp[j];

  var nr_of_zeros := i - |allSol[j]|;

  while nr_of_zeros > 0
    decreases nr_of_zeros
    invariant 0 <= nr_of_zeros <= i - |allSol[j]|
    decreases i - length
    invariant |allSol[j]| <= length <= i
    invariant |partialSolutionPrefix| == length
    invariant forall k :: 0 <= k <= length - 1 ==>
      0 <= partialSolutionPrefix[k] <= 1
    invariant length < |jobs|;
    invariant length == i - nr_of_zeros
    invariant hasNoOverlappingJobs(partialSolutionPrefix, jobs)
    invariant forall k :: j < k < |partialSolutionPrefix| ==>

```

```

    partialSolutionPrefix[k] == 0
invariant max_profit == PartialSolProfit(partialSolutionPrefix, jobs, 0)
{
    AssociativityOfProfitFunc(partialSolutionPrefix, jobs, 0, 0);
assert max_profit == PartialSolProfit(partialSolutionPrefix, jobs, 0);
    partialSolutionPrefix := partialSolutionPrefix + [0];
assert length + nr_of_zeros < |jobs|;
    length := length + 1;
    nr_of_zeros := nr_of_zeros - 1;
assert max_profit == PartialSolProfit(partialSolutionPrefix, jobs, 0);
}

assert length == i;
assert |partialSolutionPrefix| == i ;

assert forall k :: j < k < i ==> partialSolutionPrefix[k] == 0;
assert forall k :: j < k < i ==> overlappingJobs(jobs[k], jobs[i]);
assert isPartialSol(partialSolutionPrefix, jobs, i);

assert hasNoOverlappingJobs(partialSolutionPrefix + [1], jobs);

AssociativityOfProfitFunc(partialSolutionPrefix, jobs, 1, 0);
partialSolutionPrefix := partialSolutionPrefix + [1];
length := length + 1;
max_profit := max_profit + jobs[i].profit;

assert isPartialSol(partialSolutionPrefix, jobs, i + 1);
assert max_profit == PartialSolProfit(partialSolutionPrefix, jobs, 0);
forall partialSol | |partialSol| == i + 1
    && partialSolutionWithJobI(partialSol, jobs, i)
    ensures HasLessProf(partialSol, jobs, max_profit, 0)
{

    OnlyY0WhenOverlapJobs(partialSol, jobs, i, j);
assert forall k :: j < k < i ==> partialSol[k] == 0;
    ProfitLastElem(partialSol, jobs, i);

```

```

    assert PartialSolProfit(partialSol, jobs, i) == jobs[i].profit;
    OtherSolHasLessProfThenMaxProfit2
        (partialSol, jobs, i, j, max_profit, allSol, dp);
}
maxProfit := max_profit;
partialSolution := partialSolutionPrefix;
}

```

Pentru a demonstra această leamnă am presupus că ar exista o alta soluție parțială cu aceleași proprietăți care ar avea un profit mai mare decât soluția parțială optimă obținută. Astfel, în urma calculelor, am gasit că există o alta soluție parțială de lungime $j + 1$ al carui profit este mai mare decât cel al soluției parțiale optime de lungime $j + 1$, ceea ce este imposibil, deoarece ar contrazice ipoteza de la care am plecat. In acest mod, am identificat si **proprietatea de substructura optimă**, care se ascunde in această leamnă.

```

lemma HasMoreProfThanOptParSol (optimalPartialSol: seq<int>, jobs: seq<Job>,
partialSol: seq<int>)
    requires validJobsSeq(jobs)
    requires 1 <= |optimalPartialSol| <= |jobs|
    requires |optimalPartialSol| == |partialSol|
    requires isPartialSol(partialSol, jobs, |partialSol|)
    requires isOptParSol(optimalPartialSol, jobs, |optimalPartialSol|)
    requires PartialSolProfit(partialSol, jobs, 0) >
        PartialSolProfit(optimalPartialSol, jobs, 0)
    ensures !isOptParSol(optimalPartialSol, jobs, |optimalPartialSol|)
{
    var other_profit := PartialSolProfit(partialSol, jobs, 0);
    var optParSolProfit := PartialSolProfit(optimalPartialSol, jobs, 0);
    assert forall otherSol:: isPartialSol(otherSol, jobs, |optimalPartialSol|)
        ==> HasLessProf(otherSol, jobs, optParSolProfit, 0);
    assert other_profit > optParSolProfit;
    assert !isOptParSol(optimalPartialSol, jobs, |optimalPartialSol|);
}

```

În lema "HasMoreProfThanOptParSol" am confirmat că în cazul în care există o altă soluție parțială cu un profit mai mare decât cel al soluției optime, atunci soluția

considerată optimă nu poate fi cu adevărat optimă.

De asemenea, consider că și lemele pe care le-am demonstrat folosind inducția au fost la fel de complexe, precum "AssociativityOfProfitFunc".

```
lemma AssociativityOfProfitFunc (partialSolPrefix : seq<int>, jobs: seq<Job>,
val: int, index: int)
  requires 1 <= |jobs|
  requires validJobsSeq(jobs)
  requires 0 <= index <= |partialSolPrefix|
  requires 0 <= val <= 1
  requires 0 <= |partialSolPrefix| < |jobs|
  decreases |partialSolPrefix| - index
  ensures PartialSolProfit(partialSolPrefix, jobs, index)
    + val * jobs[|partialSolPrefix|].profit ==
      PartialSolProfit(partialSolPrefix + [val], jobs, index)
{
  if |partialSolPrefix| == index {

  }
  else
  {
    AssociativityOfProfitFunc(partialSolPrefix , jobs, val, index + 1);
  }
}
```

În cadrul acestei leme se poate observa cum am demonstrat proprietatea de asociativitate a funcției `PartialSolutionPrefixProfit`, care îmi calculează profitul pentru o soluție parțială, astfel încât dacă adăugam un element la capatul unei soluții parțiale, valoarea profitului acesteia crește cu valoarea profitului activității adăugate.

3.7 Mod de lucru

În timpul dezvoltării codului, un aspect important a fost modul de lucru. Pașii pe care i-am urmat pentru a-mi ușura munca au fost:

1. instrucțiunea **assume false** pentru a presupune anumite bucati de cod a fi false (Dafny nu mai încercă demonstrarea lor), încât să evit situațiile în care metodele

deveneau fragile (duc la timeout) si nu mai puteam identifica la care linie este o eroare

2. utilizarea **assert-urilor** pentru a vedea exact de la ce linie o proprietate nu mai este îndeplinită sau daca am reusit să demonstrez o anumită proprietate

3.7.1 Timeout

De-a lungul procesului de dezvoltare a codului, au existat situații în care anumite metode si leme au cauzat depășirea timpului de așteptare (timeout). Acest lucru se întâmplă atunci când acestea contin prea multe instrucțiuni care trebuie demonstrate, avand un grad de complexitate crescut. Pentru a evita timeout-urile, soluția a constat în refactorizarea metodelor si lemelor care aveau această problemă, extragand partile de cod critice pe care le-am demonstrat separat într-o alta leamnă. Prin această abordare, nu doar că am reușit să reduc timpul de demonstrare, dar am reusit și să îmbunătățesc claritatea și modularitatea codului meu, făcându-l mai ușor de întreținut și de înțeles în viitor.

Concluzii

În cadrul acestei lucrări de licență, am explorat și demonstrat corectitudinea algoritmului de selecție a activităților cu profit maxim, utilizând tehnica de proiectare cunoscută sub numele de programare dinamică. Prin intermediul limbajului imperativ Dafny, am reușit să implementez algoritmul și să verific formal corectitudinea acestuia.

Programarea dinamică s-a dovedit a fi o metodă eficientă pentru rezolvarea problemei de selecție a activităților cu profit maxim, permițându-mi să abordez problema într-un mod structurat și să optimizez soluția prin subprobleme suprapuse. Utilizarea Dafny a fost esențială pentru a garanta că implementarea respectă toate cerințele de corectitudine, prin verificarea automată a invariantilor și a proprietăților specificate.

Rezultatele obținute au confirmat faptul că algoritmul nu numai că este corect din punct de vedere formal, dar și eficient din punct de vedere al complexității timpului de execuție. Acest lucru demonstrează potențialul puternic al combinării programării dinamice cu verificarea formală folosind limbaje specializate precum Dafny în dezvoltarea algoritmilor corecți și eficienți.

Pe baza acestei lucrări, consider că există oportunități semnificative pentru cercetări viitoare, cum ar fi extinderea tehnicii pentru probleme mai complexe sau adaptarea algoritmului pentru a lucra într-un context paralel sau distribuit. De asemenea, integrarea unor alte limbaje și instrumente de verificare formală ar putea oferi perspective noi și valoroase.

În concluzie, am demonstrat că algoritmul de selecție a activităților poate fi implementat corect și eficient folosind programarea dinamică și verificarea formală în Dafny, contribuind astfel la îmbunătățirea metodelor și tehnologiilor utilizate în dezvoltarea de algoritmi siguri și fiabili.

Bibliografie

- [1] C. Andrici and Ș. Ciobâcă. Verifying the DPLL algorithm in dafny. In M. Marin and A. Craciun, editors, *Proceedings Third Symposium on Working Formal Methods, FROM 2019, Timișoara, Romania, 3-5 September 2019*, volume 303 of *EPTCS*, pages 3–15, 2019.
- [2] J. Blázquez, M. Montenegro, and C. Segura. Verification of mutable linear data structures and iterator-based algorithms in dafny. *J. Log. Algebraic Methods Program.*, 134:100875, 2023.
- [3] J. Koenig and K. R. M. Leino. Getting started with dafny: A guide. In T. Nipkow, O. Grumberg, and B. Hauptmann, editors, *Software Safety and Security - Tools for Analysis and Verification*, volume 33 of *NATO Science for Peace and Security Series - D: Information and Communication Security*, pages 152–181. IOS Press, 2012.
- [4] D. Lucanu and Ș. Ciobâcă. Lecture 11: Dynamic programming.